



HOKKAIDO UNIVERSITY

Title	マイクロ・コンピュータ用BASICプログラム Ⅰ
Author(s)	石田, 完; ISHIDA, Tamotsu
Citation	低温科学. 物理篇. 資料集, 41, 83-90
Issue Date	1983-03-10
Doc URL	https://hdl.handle.net/2115/18730
Type	departmental bulletin paper
File Information	41_p83-90.pdf



マイクロ・コンピュータ用 BASIC プログラム I *

石田 完

(低温科学研究所)

(昭和 57 年 12 月受理)

1. 最小二乗法による曲線あてはめの高速演算プログラム: Program 1

以下の説明で小文字のサフィックスは変数, 大文字のサフィックスは常数とする。精度の等しい L 個の測定値 y_i があって, これが N 個の係数 a_j の線型関数で表わせる場合,

$$a_1x_{1i} + a_2x_{2i} + \dots + a_Nx_{Ni} = y_i, \quad i = 1 \sim L$$

の L 個の連立一次方程式から N 個の係数 a_j を最小二乗法で求める。

まず文番号 1010~1050 で正規方程式の行列 A_{ij} を求める。

$$\mathbf{A} \cdot \mathbf{a} = \mathbf{y}$$

を解いて \mathbf{a} を求めればよい訳であるが, 正規方程式行列の対象性 ($A_{ji} = A_{ij}$) を利用し

$$\mathbf{A} = \mathbf{L} \cdot \mathbf{U}$$

の分解を行なう。 \mathbf{L} は \mathbf{A} の単位左下三角行列 ($L_{ii} = 1, i < j$ のとき $L_{ij} = 0$), \mathbf{U} は右上三角行列 ($i > j$ のとき $L_{ij} = 0$) である (文 1080~文 1120)。

$$\mathbf{L} \cdot \mathbf{b} = \mathbf{y}$$

を満す \mathbf{b} を求め (文 1130~文 1150)

$$\mathbf{U} \cdot \mathbf{a} = \mathbf{b}$$

を満す \mathbf{a} を求める (文 1160~文 1200)。結果は A_{im} ($M = N + 1, i = 1 \sim N$) に求まる。

ここに示した例は, 水の気化熱 L_0 を $0^\circ\text{C} \sim 100^\circ\text{C}$ で与える実験式

$$L_0 = A + B(\theta - 100) + C(\theta - 100)^2$$

の常数 A, B, C を 14 個の L_0 測定値¹⁾ から求めている。データは DATA 文で与え, X_{ij} 配列への読みこみ順序を明らかにした。

もし, $x_i, y_i, i = 1 \sim L$ が対になってフロッピーディスクに格納されている場合で,

$$a_0 + a_1x + a_2x^2 + \dots + a_nx^n = y$$

の $a_0 \sim a_n$ を求める場合なら、次のようになろう。

```

10 DEFINT I-N
20 INPUT "File name of data ? _____",F$
30 INPUT "A number of data ? _____",L
40 INPUT "A number of coefficient ? _____",N :M=N+1
50 DIM X(L,M),A(N,M)
60 OPEN F$ FOR INPUT AS #1
70 FOR I=1 TO L :INPUT #1,X(I,2),X(I,M) :NEXT
80 CLOSE #1
90 FOR I=1 TO L :X(I,1)=1
100 FOR J=2 TO N-1 :X(I,J+1)=X(I,2)^J
110 NEXT J,I

```

以下 Program 1 の文 1000 に続く。

演算を高速にするため、DEFINT I-N の文は、ぜひ必要である。算法は文献²⁾を参考にした。Program 1 のラン結果は上から 3 行が正規方程式の係数行列で、最下段が左から A , B , C の各値である。

2. 不等間隔に与えられたデータから等間隔データを作るプログラム : Program 2

不等間隔に与えられたデータのスペクトル解析を行う場合とか、点数の少ないデータを X-Y プロッタ等で滑らかな曲線に表わしたい場合に 응용して便利である。

算法は不等間隔に与えられたデータを 3 次の spline 関数によって滑らかに内挿して等間隔データを作り出す³⁾。演算の本体は Program 2 の文 100~文 270 である。

このプログラムでは $y_i = y(x_i)$, $i = 1, \dots, N$ のデータ x_i , y_i が対になって、フロッピー・ディスクに格納されていることを想定している。与えるパラメータは、元データのファイル名、元データ対の個数、等間隔に作り出す x の最小、最大値、 x の間隔である。結果は配列 $S(i)$, $i = 1, \dots, L$ (L は内部で計算される) に求まる。

本体 (文 100~文 270) を応用する場合にも

文 15 の DEFINT I-N

文 60 の $L = (UMIN - UMAX) / DU + 1$

文 70 の DIM X(N), Y(N), Z(N), S(L)

は必須である。

第 1 図は不等間隔に与えられた 11 個のデータ (A) から 16 個の等間隔データを作りだしたテスト結果 (B) で、どの程度の内挿が行なわれるかが判ると思う。

3. MEM によるスペクトル解析のプログラム : Program 3

先に MEM によるスペクトル解析プログラムを示したが⁴⁾、観測時間の長いデータを部分的に区切ってスペクトルを求め、その時間変化を見たいような場合に適したプログラムを作製したので用法を示す。

与えるパラメータ

- 1) 長いデータを区切るシリーズの数と、1シリーズ当りのデータ個数をカンマで区切ってタイプイン (文 30)。
- 2) データのはいっているファイル名 (文 40)。
- 3) 次々のシリーズを元データの何番目毎にとるかを与える。例えば第1シリーズは元データの1~100番目、第2シリーズは第1シリーズの後半をグブらせて51~150番目をとるようにするのであれば50をタイプインする (文 50)。
- 4) PEF (予測誤差フィルタ) の項数 (文 60)。
- 5) 求めるスペクトルの最小と最大周波数。Hz で与えるよう指示しているが、無次元周波数なら最大範囲は0と0.5である (文 70)。
- 6) 求めるスペクトルの分解巾とデータの間隔。Hz と Sec で与えるように指示しているが、無次元周波数なら分解巾は求めるスペクトルの項数から計算して与え、データ間隔は1とする (文 80)。
- 7) 結果を格納するファイル名 (文 90)。

計算時間を短縮するため、自己相関関数の計算ははぶいてある。しかし FPE (予測誤差の期待値) とスペクトル値は表示される。もちろん1シリーズのデータでも計算できる。その場合は1) でシリーズ数は1を、3) では0を与えればよい。

4. 度数分布をプリントするプログラムの訂正: Program 4

先の Program 13⁴⁾ で一部虫があった。つまり、データの最大値が0か負であった場合、エラーを生ずるので元の文番号150を新 Program 4の如く、文150及び文155の如く変更した。なお新プログラムでは、度数のクラス巾を任意に与えられるようにした。すなわち文65で自動的に巾を定めない場合にはNをタイプインし、文67でその巾の値をタイプインする。

Program 4の後に、テスト・データを用いた実行結果と、そのテスト・データの作製プログラムを示した。

文 献

- 1) 芝亀吉 1943 最小自乗法、応用数学第9巻、河出書房、96-99
- 2) 伊理正夫 1982 数値計算の常識、5. bit, 14, No. 9, 26-27
- 3) 斉藤正徳 1974 時系列解析用のサブルーチン、SEISDAP (Seismic Data Processing Subroutine Package)、東京大学大型計算機センター。
- 4) 石田完 1981 時系列データ解析のための BASIC プログラム、低温科学、物理篇、資料集、40, 87-113

Program 1

```

1  '***   Data       ***
10 DATA 3,14
11 DATA 1,-100,10000,2500.71
12 DATA 1, -95, 9025,2488.32
13 DATA 1, -90, 8100,2476.71
14 DATA 1, -85, 7225,2464.78
15 DATA 1, -80, 6400,2453.32
16 DATA 1, -75, 5625,2441.25
17 DATA 1, -70, 4900,2429.36
18 DATA 1, -60, 3600,2405.75
19 DATA 1, -50, 2500,2381.65
20 DATA 1, -40, 1600,2357.34
21 DATA 1, -30, 900,2332.93
22 DATA 1, -20, 400,2307.81
23 DATA 1, -10, 100,2282.45
24 DATA 1, 0, 0,2256.32
100 '*** Input of data ***
110 DEFINT I-N : READ N,L : M=N+1
120 DIM X(L,M),A(N,M)
130 FOR I=1 TO L:FOR J=1 TO M
140 READ X(I,J):NEXT J,I
1000 '*** Standard equation ***
1010 FOR J=1 TO N:FOR K=1 TO M
1020 IF K<J THEN A(J,K)=A(K,J):GOTO 1050
1030 S=0:FOR I=1 TO L:S=S+X(I,J)*X(I,K):NEXT
1040 A(J,K)=S
1050 NEXT K,J
1060 GOSUB 1230
1070 '*** Doolittle-method ***
1080 FOR K=1 TO N-1:W=1/A(K,K)
1090 FOR I=K+1 TO N:A(I,K)=W*A(I,K):NEXT I
1100 FOR I=K+1 TO N:FOR J=K+1 TO N
1110 A(I,J)=A(I,J)-A(I,K)*A(K,J)
1120 NEXT J,I,K
1130 FOR K=1 TO N-1:FOR I=K+1 TO N
1140 A(I,M)=A(I,M)-A(I,K)*A(K,M)
1150 NEXT I,K
1160 FOR K=1 TO N:W=1/A(K,K)
1170 FOR J=K+1 TO M:A(K,J)=W*A(K,J):NEXT
1180 FOR I=1 TO K-1
1190 FOR J=K+1 TO M:A(I,J)=A(I,J)-A(I,K)*A(K,J):NEXT
1200 NEXT I:NEXT K
1210 FOR I=1 TO N:PRINT A(I,M),:NEXT
1220 END
1230 '*** Print of array ***
1240 FOR I=1 TO N:FOR J=1 TO M
1250 PRINT A(I,J),:NEXT
1260 PRINT :NEXT :PRINT
1270 RETURN

RUN
  14          -805          60375          33578.7
-805          60375          -4.91838E+06      -1.96498E+06
 60375          -4.91838E+06      4.18622E+08      1.48308E+08

2256.85       -2.5587         -1.27527E-03


```

Ok

Program 2

```

10 '*** Equal interval-Spline interpolation ***
15 DEFINT I-N
20 INPUT"File name of X,Y Pair data ?___ ",F$
30 INPUT"A number of data-Pair ?----- ",N
40 INPUT"Min and Max values of new X ?___ ",UMIN,UMAX
50 INPUT"Interval of new X ?----- ",DU
60 L=(UMAX-UMIN)/DU+1
70 DIM X(N),Y(N),Z(N),S(L)
80 OPEN F$ FOR INPUT AS #1
90 FOR I=1 TO N :INPUT #1,X(I),Y(I) :NEXT :CLOSE #1
100 Z(1)=0 :Z(N)=0 :S(N)=0
110 IF N=2 GOTO 190
120 FOR J=N-1 TO 2 STEP -1
130 H1=X(J)-X(J-1) :H2=X(J+1)-X(J) :H3=H1+H2
140 CL=H2/H3 :CM=H1/H3
150 CD=(CM*Y(J+1)-Y(J)+CL*Y(J-1))/(H1*H2)
160 CE=1/(2+CL*S(J+1)) :Z(J)=(CD-CL*Z(J+1))*CE
170 S(J)=-CM*CE :NEXT
180 FOR J=1 TO N-2 :Z(J+1)=Z(J+1)+S(J+1)*Z(J) :NEXT
190 J=1 :FOR I=1 TO L :U=UMIN+DU*(I-1) :SS=0
200 IF U<X(1) OR U>X(N) GOTO 270
210 ON SGN(U-X(J))+2 GOTO 250,240,220
220 J=J+1
230 IF J>N THEN 270 ELSE 210
240 SS=Y(J) :GOTO 270
250 H1=X(J)-X(J-1) :D1=U-X(J-1) :D2=X(J)-U
260 SS=(D2*Y(J-1)+D1*Y(J)-D1*D2*(Z(J-1)*(H1+D2)+Z(J)*(H1+D1)))/H1
270 S(I)=SS :NEXT :PRINT :PRINT "The result has";L;" terms."
280 BEEP :PRINT :INPUT"File name of result ?----- ",F$
290 OPEN F$ FOR OUTPUT AS #1
300 FOR I=1 TO L :PRINT #1,S(I) :NEXT :CLOSE #1
310 END
    
```



第1図

Program 3

```

10 '*** MEM-analysis of series data ***
20 DEFINT I-N
30 INPUT"Number of series and data Per a series ";NSR,N:NSR1=NSR-1
40 INPUT"File name of data ";FIN$
50 INPUT"La9 number of series ";LSR
60 INPUT"Term number of PEF ";LAG:LGM1=LAG-1
70 INPUT"Min and max frequency (Hz) ";FMIN,FMAX
80 INPUT"Band width (Hz) and data interval (s) ";DF,DT
90 INPUT"File name of results ";FOT$
100 FFN=FMIN*DT : FFX=FMAX*DT
110 IF FFX>.5 THEN FFX=.5
120 DFF=DF*DT : KK=INT((FFX-FFN)/DFF+1.5) : K1=KK-1
130 PI#=6.283185307179599# : NP1=N+1:NM1=N-1
140 DIM X(N),Y(NSR1,K1),S(KK),A(LAG),B1(N),B2(N),G(LAG)
150 CLS
160 FOR ISR=0 TO NSR1
170 OPEN FIN$ FOR INPUT AS #1
180 FOR IXD=1 TO LSR*ISR : INPUT #1,XD : NEXT
190 FOR IDT=1 TO N : INPUT #1,X(IDT) : NEXT : CLOSE #1
200 SX=0 : FOR I=1 TO N : SX=SX+X(I) : NEXT
210 XM=SX/N : FOR I=1 TO N : X(I)=X(I)-XM : NEXT
220 GOSUB 300
230 FOR I=1 TO KK : J=I-1 : Y(ISR,J)=S(I) : NEXT
240 CLS 2,1:LOCATE 0,2:PRINT "ISR=";ISR
250 NEXT ISR
260 OPEN FOT$ FOR OUTPUT AS #1
270 FOR I=0 TO NSR1 : FOR J=0 TO K1 : PRINT #1,Y(I,J);:NEXT J,I
280 CLOSE #1 :END
290 '*** SUB MEM ***
300 SUM=0 : FOR I=1 TO N : SUM=SUM+X(I)^2 : NEXT
310 PM=SUM/N : B1(1)=X(1) : FPE=NP1/NM1*PM
320 FOR I=2 TO N : B1(I)=X(I) : B2(I-1)=X(I) : NEXT
330 FOR M=1 TO LGM1
335 BEEP:CLS 0,1:PRINT "Go on G(";M;") !","FPE=";FPE
340 STN=0 : STD=0 : IM=N-M
350 FOR I=1 TO IM : STN=STN+B1(I)*B2(I):ST0=STD+B1(I)^2+B2(I)^2 : NEXT
360 A(M)=2*STN/STD : PM=PM*(1-A(M)^2)
370 IF M=1 THEN 390
380 MM1=M-1 : FOR K=1 TO MM1 : A(K)=G(K)-A(M)*G(M-K) : NEXT
390 FOR I=1 TO M : G(I)=A(I) : NEXT : IMM=N-M-1
400 FOR I=1 TO IMM : B1(I)=B1(I)-G(M)*B2(I):B2(I)=B2(I+1)-G(M)*B1(I+1) : NEXT
410 FPE=(NP1+M)/(NM1-M)*PM
420 NEXT M : G(1)=1 : PMDT=PM*DT
425 BEEP:CLS 0,1:PRINT "Go on G(";M;") !","FRE=";FPE
430 FOR J=2 TO LAG : G(J)=-A(J-1) : NEXT
440 FOR I=1 TO KK : FR=FFN+DFF*(I-1) : W#=FR*PI#
450 CW#=COS(W#) : SW#=SIN(W#) : CW2#=2*CW# : UV#=0 : UVU#=0 : IZ=LAG
460 FOR IQ=1 TO LGM1 : UVU#=UVU# : UVU#=UV# : UV#=G(IZ)+UVU#*CW2#-UVU# : IZ=IZ-1 : NEXT
470 CC=G(1)+UV#*CW#-UVU# : SS=UV#*SW# : CS=CC^2+SS^2
480 S(I)=PMDT/CS
490 BEEP : CLS 0,1: PRINT "Go on S(";I;")=";S(I)
500 NEXT I : RETURN

```

Program 4

```

10 '*** HINDO ***
20 DEFINT I,X,Y,Z,K,N
30 INPUT "File name of data ";FIL$
40 INPUT "Number of data ";N
50 INPUT "Name of data ";F$
60 INPUT "Name of unit ";U$
65 INPUT "Auto-step width (Y/N) ";W$
66 IF W$="Y" THEN 70
67 INPUT "Step width ";W
70 N=N-1 : DIM A(N)
80 OPEN FIL$ FOR INPUT AS #1
90 FOR I=0 TO N : INPUT #1,A(I) : NEXT
100 '*** Calculation ***
110 IF A(0)>A(1) THEN AMAX=A(0):AMIN=A(1) ELSE AMAX=A(1):AMIN=A(0)
120 FOR I=2 TO N
130 IF A(I)>AMAX THEN AMAX=A(I) ELSE IF A(I)<AMIN THEN AMIN=A(I)
140 NEXT I
150 AA=AMAX : IF AA<=0 THEN AA=ABS(AMIN)
155 GOSUB 600 : IF F2=0 THEN 175
160 FOR I=0 TO N : A(I)=A(I)*F1 : NEXT I
170 AMAX=AMAX*F1 : AMIN=AMIN*F1
175 IF W$="Y" THEN 180
176 W=W*F1 : W2=W*.5 : GOTO 190
180 W=20*(AMAX-AMIN)/N+.05 : W=INT(W*10)/10 : W2=W/2
190 Y=INT(AMIN/W)+1 : Z=INT(AMAX/W)+1 : IE=Z-Y
200 DIM K(IE),P(IE)
210 M21=0 : M22=0 : M23=0
220 FOR I=0 TO IE : K(I)=0 : NEXT
230 FOR I=0 TO N : X=INT(A(I)/W)+1-Y : K(X)=K(X)+1 : NEXT : IMAX=1
240 KMAX=K(0) : FOR I=1 TO IE : IF KMAX<K(I) THEN KMAX=K(I):IMAX=I:NEXT
250 KSUM=0 : FOR I=0 TO IE : KSUM=KSUM+K(I) : NEXT
260 FOR I=0 TO IE : P(I)=K(I)/KSUM*100 : NEXT
270 FOR I=0 TO IE : II=I-IMAX
280 M21=M21+II*K(I) : M22=M22+II^2*K(I) : M23=M23+II^3*K(I)
290 NEXT
300 M21=M21*W/KSUM : M=M21+(IMAX+Y)*W-W2 : M22=M22*W^2/KSUM
310 M23=M23*W^3/KSUM : M2=M22-M21^2 : RMS=SQR(M2)
320 M3=M23-3*M22*M21+2*M21^3 : SK=M3/RMS^3
330 '*** Print of result ***
340 DEF CHR$(%HA5)="000000C0C0000000"
350 D$=CHR$(%HA5) : C$=CHR$(%HF1) : S$=STRING$(9," ")
360 LPRINT "Frequency distribution of ";F$
370 LPRINT:LPRINT "Unit=";:LPRINT 10^F2 ;:LPRINT U$ : LPRINT SPC(11)
380 LPRINT "0      10      20      30      40      50%"
390 LPRINT USING "MIN:###.## ";AMIN;
400 LPRINT " |          |          |          |          | "
410 FOR I=0 TO IE
420 U=W*(I+Y)-W2 : IP=INT(P(I)+.5) : U$=STRING$(IP,CHR$(%HA0))
430 LPRINT USING "    ###.## ";U;
440 IF IP< 1 THEN LPRINT C$+S$+D$+S$+D$+S$+D$+S$+D$+S$+D$+S$+D$ SPC(4);:GOTO 520
450 IF IP<= 9 THEN LPRINT U$SPC(10-IP)D$+S$+D$+S$+D$+S$+D$+S$+D$+S$+D$SPC(4);:GOTO 520
460 IF IP<=19 THEN LPRINT U$SPC(20-IP)D$+S$+D$+S$+D$+S$+D$+S$+D$SPC(4);:GOTO 520
470 IF IP<=29 THEN LPRINT U$SPC(30-IP)D$+S$+D$+S$+D$+S$+D$SPC(4);:GOTO 520
480 IF IP<=39 THEN LPRINT U$SPC(40-IP)D$+S$+D$+S$+D$SPC(4);:GOTO 520
490 IF IP<=49 THEN LPRINT U$SPC(50-IP)D$SPC(4);:GOTO 520
500 IF IP =50 THEN LPRINT U$SPC(5);:GOTO 520
510 IF IP> 50 THEN LPRINT STRING$(50,CHR$(%HA0))+ "Over ";
520 LPRINT USING "###.##";P(I); : LPRINT USING "    ####";K(I)
530 NEXT I
540 LPRINT USING "MAX:###.## ";AMAX;
550 LPRINT " |          |          |          |          | "
560 LPRINT TAB(69)"Total=";KSUM
570 LPRINT : LPRINT TAB(11)"Mean=";M*10^F2;U$;" RMS=";RMS*10^F2;U$;" Skewness=";SK
580 END
600 '*** SUB F1,F2 ***
610 F2=0
620 IF AA>=100 THEN 650 ELSE IF AA>=10 THEN 670 ELSE 630
630 AA=AA*10 : F2=F2-1
640 IF AA<10 THEN 630 ELSE 670
650 AA=AA/10 : F2=F2+1

```

Program 4 の続き

```

660 IF AA>=100 THEN 650 ELSE 670
670 F1=10^-F2
680 RETURN

```

```

RUN

```

```

File name of data ? 2:TTT.DAT
Number of data ? 100
Name of data ? Test data
Name of unit ? degree
Auto-step width (Y/N) ? N
Step width ? 20
Frequency distribution of Test data

```

```

Unit= 10 degree

```

	0	10	20	30	40	50%		
MIN:-12.00								
-11.00	■	1.0	1
-9.00	■	6.0	6
-7.00	■	■	25.0	25
-5.00	■	■	■	.	.	.	40.0	40
-3.00	■	■	■	.	.	.	20.0	20
-1.00	■	5.0	5
1.00	■	3.0	3
MAX: 0.00								

Total= 100

Mean=-50.2 degree RMS= 22.3598 degree Skewness= .320242

```

Ok

```

```

10 '*** TTT.DAT ***
20 DIM A(100) : A(1)=-120
30 FOR I=2 TO 7 : A(I)=-100
40 FOR I=8 TO 32 : A(I)=-80
50 FOR I=33 TO 72 : A(I)=-60
60 FOR I=73 TO 92 : A(I)=-40
70 FOR I=93 TO 97 : A(I)=-20
80 FOR I=98 TO 100 : A(I)=0
90 OPEN "2:TTT.DAT" FOR OUTPUT AS #1
100 FOR I=1 TO 100 : PRINT #1,A(I); :NEXT
110 CLOSE #1 : END

```